Gross-Pitaevskii

- Spherical Gross-Pitaevskii.
- Imaginary time.
- Fortran 77 program.
- Example for N=100000.

$$\left[-\frac{1}{2}\vec{\nabla}^2 + \frac{1}{2}r^2 + 4\pi \bar{a}_s N |\psi|^2 \right] \psi = \mu \psi$$

$$e = \frac{E}{N} = \int d^3r \; \psi^* \left[-\frac{1}{2}\vec{\nabla}^2 + \frac{1}{2}r^2 \right] \psi + 2\pi \bar{a}_s N \int d^3r \, |\psi|^4$$

$$R(0) = 0$$

$$\psi = \frac{R}{r} Y_{00} \qquad \psi \text{ normalized to } 1$$

$$\left[-\frac{1}{2}\frac{d^2}{dr^2} + \frac{1}{2}r^2 + 4\pi \bar{a}_s N \left(\frac{R}{r}\right)^2 \frac{1}{4\pi} \right] R(r) = \mu \, R(r)$$

$$U = e^{-\frac{itH}{\hbar}} \xrightarrow{it = \tau} U = e^{-\frac{\tau H}{\hbar}}$$

$$t = -i\tau$$

We take a trial wave function at $\tau = 0$, with a non-zero overlap with the g.s.

$$\Psi(x, \tau = 0) = \sum_n c_n \varphi_n(x)$$

$$\psi(x,t) = \sum_n c_n \varphi_n(x)\, e^{-E_n t/\hbar}$$

$$\downarrow\ t \to \infty$$

$$\psi(x,t) \longrightarrow c_0 \varphi_0(x)\, e^{-E_0 t/\hbar}$$

The states different than φ_0 decay faster and only survives the g.s.

How we do the evolution if we do not know the eigenstates?

$$\psi(x, t+\Delta t) = \psi(x, t) - \frac{\Delta t\, H}{\hbar}\, \psi(x, t)$$

Δt very small

We loose normalization $\Rightarrow$ at each time step we normalize the wave-function

$$R(r, t+\Delta t) = R(r,t) - \Delta t\, H\, R(r,t)$$

$$e = \int dr \left[-\frac{1}{2} R R'' + \frac{1}{2} R^2 r^2 \right] + \frac{\bar{a}_s N}{2} \int dr\, r^2 \left(\frac{R}{r}\right)^4$$

$$\langle K \rangle = -\int dr \frac{1}{2} R R''$$

$$\langle H_0 \rangle = \frac{1}{2} \int dr\, R^2 r^2$$

$$\langle int \rangle = \frac{\bar{a}_s N}{2} \int dr\, r^2 \left(\frac{R}{r}\right)^4$$

$$\langle pot \rangle = \langle H_0 \rangle + \langle V_{int} \rangle$$

$$u(r) = \frac{1}{2} r^2 + \bar{a}_s N \left(\frac{R}{r}\right)^2$$

$$\begin{cases} \rho(r) = \left(\frac{R}{r}\right)^2 \frac{1}{4\pi} \\ \int d^3r\, \rho(r) = 1 \end{cases}$$

Radius: $\sqrt{\int dr\, r^2 R^2}$

Average μ: $\int dr\, \mu(r) R^2(r)$

Virial: $2 E_{kin} - 2 E_{pot}(k,0) + 3 E_{pot}(int) = 0$

```
      PROGRAM MAIN
      IMPLICIT REAL*8(A-H,O-Z)
      dimension xr(1000),frev(1000),freo(1000),fred(1000),xmu(1000),
     1  fren(1000),den(1000),u(1000)
C     AA=10000. NUmber of atoms
c     N1 numbre of integration steps in r-grid
c     step, integration step in r-space
c     a0 scattering length in h.o. units
c     alpha , the starting parameter for the h.o. function
c     time , the step in time
c     iter, number of iterations
      read(5,*)a0,n1,step,aa,time,alpha,iter
 340  format(2d15.5)
      WRITE(6,1000)AA,N1,step,a0,ALPHA,time,iter
 1000 FORMAT(/5X,'*',F10.0,' BOSONS IN A SPHERICAL TRAP *'/
     1 5X,'* r-GRID IN N1=',I4,' POINTS', ' r-step',f8.3/
     1 5X,'* A0=',E12.4,2X,' ALPHA=',E12.4/
     1 5x,'* time=',e12.4,' number-iter =',i8/)
      pi=4.d0*DATAN(1.d0)
      piin=1.d0/(4.d0*pi)
      pi2in=dsqrt(piin)
      alpha2=alpha*alpha
      cvar=2.d0*dsqrt(alpha)**3/dsqrt(dsqrt(pi))
c     building the starting wave function R(r). Phi(r)=R(r)/r*Y00
      do i=1,n1
      xr(i)=step*dfloat(i-1)
      xr2=xr(i)*xr(i)
      frev(i)=cvar*xr(i)*dexp(-0.5d0*alpha2*xr2)
      freo(i)=frev(i)
      enddo
c     starting the convergence process
****************************************
c to reduce to the armonic oscillator cequ=0
************************************
      cequ=a0*aa
c     if you put cequ=0 you recover the h.o.
      as3n=aa*a0*a0*a0
c     cequ=0.d0
      itw=0
      do 2340 it=1,iter
      itw=itw+1
 500     xnorm=0.d0
      ene0=0.d0
```

```
      cequ=a0*aa
:     if you put cequ=0 you recover the h.o.
      as3n=aa*a0*a0*a0
:     cequ=0.d0
      itw=0
      do 2340 it=1,iter
      itw=itw+1
500      xnorm=0.d0
      ene0=0.d0

      do i=2,n1-1
      fred(i)=(freo(i-1)+freo(i+1)-2.d0*freo(i))/(step*step)
      enddo
      fred(n1)=(freo(n1-1)-2.d0*freo(i))/(step*step)
      do i=1,n1
      xr2=xr(i)*xr(i)
      if(i .eq. 1)then
      xmu(i)=0.d0
      else
      ene0=ene0-freo(i)*fred(i)*0.5d0
     1 +0.5d0*xr2*freo(i)*freo(i)
     1 +0.5d0*cequ*xr2*(freo(i)/xr(i))**4
      xmu(i)=-0.5d0*fred(i)/freo(i)
     1 +0.5d0*xr2+cequ*(freo(i)/xr(i))**2
      endif
      fren(i)=freo(i)-time*xmu(i)*freo(i)
      xnorm=xnorm+fren(i)*fren(i)
      enddo
```

```
      xnorm=dsqrt(xnorm*step)
      ene0=ene0*step
      if (itw .eq. 200)then
      write(6,*) 'ene0=',ene0
      itw=0
      endif
c     I define the new wf.
      do i=1,n1
      freo(i)=fren(i)/xnorm
      enddo
      if ( it .eq. iter)then
      write(10,340)(xr(i),xmu(i),i=2,n1)
      else
      endif
 2340 continue
c     calculation ofthe radious, potential and kinetic energy, density and
c     single particle potential
      do i=2,n1-1
      fred(i)=(freo(i-1)+freo(i+1)-2.d0*freo(i))/(step*step)
      enddo
      fred(n1)=(freo(n1-1)-2.d0*freo(i))/(step*step)

      radious=0.d0
      xkin=0.d0
      potho=0.d0
      potself=0.d0
      chem=0.d0
      xaver=0.d0
      xnormden=0.d0
```

```
      do i=2,n1
      xr2=xr(i)*xr(i)
      radious=radious+xr2*freo(i)*freo(i)
      xkin=xkin+freo(i)*fred(i)
      poth0=poth0+xr2*freo(i)*freo(i)
      potself=potself+xr2*(freo(i)/xr(i))**4
      chem=chem+xmu(i)*freo(i)*freo(i)
      u(i)=0.5d0*xr2+cequ*(freo(i)/xr(i))**2
      den(i)=(freo(i)/xr(i))**2
      xnormden=xnormden+den(i)*xr2
      xaver=xaver+freo(i)*freo(i)*as3n*den(i)
      enddo
      radious2=radious*step
      radious=dsqrt(radious*step)
      xaver=xaver*step
      chem=chem*step
      xkin=-xkin*step*0.5D0
      poth0=0.5d0*poth0*step
      potself=potself*step*cequ*0.5d0
      pot=potself+poth0
      xnormden=xnormden*step
      write(6,*) ' xnormden = ', xnormden
      write(6,740)ene0,chem,xkin,pot,poth0,potself,radious,radious2
  740 FORMAT(/5X,'* ener',e12.5, '  average chemical=', e12.5/
     1 5X,'* kin-ener=',e12.5,'   total-pot= ', e12.5,/
     1 5X,'* potho=',E12.4,2X,'   potint =',E12.4/
     1 5x,'* radious =',e12.4, 5x, ' radious2 =',e15.7, /)
      do i=2,n1
      write(9,'(2e15.5)') xr(i),den(i)
      enddo
  345 continue
      stop
      end
```

22 mar 17 12:06 **out**

```
      *     100000. BOSONS IN A SPHERICAL TRAP *
      * r-GRID IN N1= 600 POINTS r-step    0.015
      * A0=  0.4330E-02   ALPHA=  0.4000E+00
      * time=  0.1000E-03  number-iter =    50000

ene0 =  12.1815034
ene0 =  12.1135559
ene0 =  12.1056062
ene0 =  12.1042696
ene0 =  12.1040149
ene0 =  12.1039637
ene0 =  12.103953
ene0 =  12.1039506
ene0 =  12.1039501
ene0 =  12.10395
ene0 =  12.1039499
ene0 =  12.1039499
ene0 =  12.1039499
ene0 =  12.1039499
ene0 =  12.1039499
ene0 =  12.1039499
ene0 =  12.1039499
```

```
normden =    1.

  * ener 0.12104E+02    average chemical= 0.16847E+02
  * kin-ener= 0.12367E+00      total-pot=  0.11980E+02
  * potho=  0.7238E+01        potint =  0.4743E+01
  * radious =  0.3805E+01       radious2 =  0.1447527E+02
```


Density profile ($a_s/a_{h.o.}$=0.00433)
Units h.o.
N=100000
rho(r)
r
0,04
0,03
0,02
0,01
0
0
1
2
3
4
5
6
7

TF

$\bar{a}_s = 0.00433$

$N = 100000$

$$\mu = \frac{1}{2}\left(15\,\bar{a}_s\,N\right)^{2/5} = 16.75$$

$$e = \frac{5}{7}\mu = 11.96$$

$$\langle r^2 \rangle = \frac{6}{7}\mu = 14.36$$